1 予備知識

2 次方程式 $a x^2 + b x + c = 0$ の解は、公式 $x = \dfrac{-b \pm \sqrt{b^2 - 4 a c}}{2 a}$ により与えられることが知られています。この公式は、平方完成と呼ばれる操作から得られます。具体的には、元の 2 次式を

$$\begin{aligned} a x^2 + b x + c &= a \left(x^2 + \frac{b}{a} x \right) + c \\ &= a \left(x + \frac{b}{2 a} \right)^2 + c - a \left(\frac{b}{2 a} \right)^2 \qquad \text{←この変形がちょっと天下り的} \end{aligned}$$

のように、1 次の項を除去した形 $a (x + p)^2 + q = 0$ に変形することにより、

$$\begin{aligned} a (x + p)^2 + q &= 0 \\ (x + p)^2 &= -\frac{q}{a} \\ x + p &= \pm \sqrt{-\frac{q}{a}} \\ x &= -p \pm \sqrt{-\frac{q}{a}} \end{aligned}$$

解の公式が導かれます[*1]。

3 次や 4 次の方程式の場合、解の公式の導出には、2 次式における平方完成に相当する「2 番目の項[*2]を除去する変形」

$$\begin{aligned} a x^3 + b x^2 + c x + d = 0 &\quad \rightharpoonup \quad a X^3 + p X + q = 0 \\ a x^4 + b x^3 + c x^2 + d x + e = 0 &\quad \rightharpoonup \quad a X^4 + p X^2 + q X + r = 0 \end{aligned}$$

が最初のステップとして用いられます。ここで、元の変数 x と変形後の変数 X との間の関係式（変数変換）は、3 次式の場合は $X = x + \dfrac{b}{3 a}$、4 次式の場合は $X = x + \dfrac{b}{4 a}$ です。例えば、3 次方程式 $X^3 + p X + q = 0$（2 次の項が無い）であれば、$X = T - \dfrac{p}{3 T}$ と置くことにより、

$$\begin{aligned} \left(T - \frac{p}{3 T} \right)^3 + p \left(T - \frac{p}{3 T} \right) + q &= 0 \\ T^3 + q - \frac{p^3}{27 T^3} &= 0 \\ T^6 + q T^3 - \frac{p^3}{27} &= 0 \end{aligned}$$

[*1] 最後の式 $x = -p \pm \sqrt{-q/a}$ に $p = b/(2 a)$、$q = c - a \{b/(2 a)\}^2$ を代入すると、係数 a、b、c に関する解の公式 $x = \{-b \pm \sqrt{b^2 - 4 a c}\}/(2 a)$ が得られます。

[*2] 一般に、n 次式における $n - 1$ 次の項を「2 番目の項」と呼ぶことにしています。

と変形できることから、x の値を求めることが出来ます。

2 Maxima による実装

多項式を引数にとり、「2 番目の項を除去する」関数 `elim_2nd` を Maxima を使って定義してみましょう。

– elim_2nd.mac –

```
elim_2nd([L]) := block([f, F, x, p, q, h, X, V,
    list: []],
    f: expand(L[1]),
    if length(L) = 2 then (
        x: L[2],
        h: hipow(f, x),
        if h >= 2 then (
            p: factor(coeff(f, x, h - 1)/(h * coeff(f, x, h)
                )),
            F: collectterms(expand(subst(X - p, x, f)), X),
            return(subst(x + p, X, F))
        )
    ) else (
        V: listofvars(f),
        for x in V do (
            h: hipow(f, x),
            if h >= 2 then (
                p: factor(coeff(f, x, h - 1)/(h * coeff(f, x
                    , h))),
                F: collectterms(expand(subst(X - p, x, f)),
                    X),
                list: cons(subst(x + p, X, F), list)
            )
        ),
        return(list)
    )
);
```

$(x+k)^h$ を展開すると、$x^h + h\,k\,x^{h-1} + \cdots$ となることから、h 次式 $a\,x^h + b\,x^{h-1} + \cdots$ を「2 番目の項を除去した形」$a\,X^h + p\,X^{h-2} + \cdots$ に変形するには、

$$a\,(x+k)^h = a\,x^h + a\,h\,k\,x^{h-1} + \cdots \quad と \quad a\,x^h + b\,x^{h-1} + \cdots$$

の係数比較により、$X = x + \dfrac{b}{a\,h}$ と置き換えればよいことがわかります[*3]。そこで、変数 x に $X - \dfrac{b}{h\,a}$ を代入し、変数 X の多項式として整理しておき（9 行目、18 行目）、最後に変数を x に戻して出力します（10 行目）。

ここで定義した関数 elim_2nd は、多項式と変数を引数に与えて実行する標準的な利用方法の他、変数を省略し多項式のみを引数に与えて実行することも出来るように定義しています。変数を省略した場合は、多項式に用いられている全ての文字（13 行目）を変数と見なし、(次数が 2 以上なら)「2 番目の項を除去した形」に変形し（リスト list に蓄積し）、(最後にまとめて）出力します（22 行目）。

実行例は次の通りです。

```
(%i1) load("elim_2nd.mac");

(%o1)                            elim_2nd.mac

(%i2) elim_2nd(a*x^2 + b*x + c, x);
                                                  2
                                     b   2       b
(%o2)                        a (x + ---)  + c - ---
                                    2 a         4 a

(%i3) elim_2nd(a*x^3 + b*x^2 + c*x + d, x);
                                          2                     3
                      b   3          b          b         b c   2 b
(%o3)           a (x + ---)  + (c - ---) (x + ---) + d - --- + -----
                      3 a            3 a       3 a       3 a      2
                                                              27 a

(%i4) elim_2nd(x^2 + a^2*x + a*c);
                        2            4                         2
                       a   2        a     2      c      2     c
(%o4)            [(x + --)  + a c - --, x  + (--- + a)  x - ---]
                       2            4           2 x           4 x
```

最後の実行例 (%i4) では、変数 x についての平方完成 $\left(x + \dfrac{a^2}{2}\right)^2 + a\,c - \dfrac{a^4}{4}$ と変数 a についての平方完成 $x\left(a + \dfrac{c}{2\,x}\right)^2 + x^2 - \dfrac{c^2}{4\,x}$ が出力されています。

作成日：平成 21 年 8 月 30 日
ソフトウェア：Maxima 5.19post & Clozure Common Lisp Version 1.3-r11936

[*3] $a\,h\,k = b$ を k について解くと、$k = b/(a\,h)$ となります。